OneCo Implementation Showcase

http://drorbn.net/AcademicPensieve/Projects/OneCo-1604/
Started 2016-04-12; modified 2016-04-13 11:02:09-04:00

In $\mathcal{U}(T)\otimes\mathcal{U}(H)$ conventions.

SetSigns.

```
Cases[SetSigns[5], ϵ_, ∞] // Union
```

```
{ϵ5, ϵ6, ϵ9, ϵ10}
```

```
Table[i → ϵi, {i, 0, 48}]
```

```
{0 → 1, 1 → 1, 2 → ϵ10, 3 → ϵ5, 4 → ϵ10, 5 → ϵ5,
 6 → ϵ6, 7 → -ϵ5, 8 → ϵ5 ϵ9, 9 → ϵ9, 10 → ϵ10, 11 → ϵ10,
 12 → ϵ10, 13 → -ϵ10, 14 → ϵ10, 15 → ϵ10, 16 → -ϵ9 ϵ10,
 17 → ϵ9 ϵ10, 18 → ϵ5 ϵ10/ϵ6, 19 → ϵ5, 20 → ϵ5, 21 → ϵ5,
 22 → ϵ5, 23 → ϵ10, 24 → ϵ10, 25 → ϵ10, 26 → ϵ10, 27 → ϵ10,
 28 → ϵ10, 29 → ϵ10, 30 → ϵ5 ϵ10/ϵ6, 31 → ϵ5, 32 → ϵ5 ϵ10,
 33 → ϵ5 ϵ10, 34 → ϵ5 ϵ10, 35 → ϵ10, 36 → ϵ5, 37 → ϵ10,
 38 → ϵ10, 39 → ϵ10, 40 → ϵ10, 41 → ϵ10^2, 42 → ϵ10,
 43 → ϵ10, 44 → ϵ10^2, 45 → ϵ10, 46 → ϵ10, 47 → ϵ10, 48 → ϵ10}
```

Generalities.

```
DQ[is___] := (Sort[{is}] === Union[{is}]);
OQ[is___] := OrderedQ[{is}];
(*Also true if {is}={i,i}*)
Kδ /: Kδ_{is__} := KroneckerDelta[1, Length[Union[{is}]]];
Simp[expr_] := Expand[expr];
S[expr_] :=
  expr /. (λ_β | λ_a | λ_δβ | λ_δa | λ_c | λ_ca | λ_δaa) :→
    MapAt[Simp, λ, 1];
AutoCollecting[λ_] := (λ /: λ[0, ___] = 0;
   λ /: λ[f_, r___] + λ[g_, r___] := λ[Simp[f + g], r];
   λ /: g_ * λ[f_, r___] := λ[Simp[g f], r]);
AutoCollecting /@ {β, a, δβ, c, δa, ca, δaa};
UU /: UU[x_] + UU[y_] := UU[x + y];
UU /: a_ * UU[x_] := UU[Expand[a x]];
γ[f_, j_, k_] := δa[f, j, k] - c[ϵ0 b_j f, k];
γa[f_, j_, k_, l_, m_] :=
  δaa[f, j, k, l, m] - ca[ϵ0 b_j f, k, l, m];
```

δaa relations.

```
UU[expr_] // S := UU[S[expr //. {
      δaa[f_, i_, j_, k_, l_] /; ! OQ[j, l] :→
        δaa[f, k, l, i, j],
      δaa[f_, i_, j_, k_, l_] /;
         ! OQ[i, k] ⋀ DQ[j, l] ⋀ OQ[j, l] :→
        δaa[f, i, l, k, j] + ca[ϵ1 b_k f, l, i, j] +
         ca[-ϵ1 b_i f, l, k, j] + ca[-ϵ1 b_k f, j, i, l] +
         ca[ϵ1 b_i f, j, k, l],
      δaa[f_, i_, k_, j_, k_] /; ! OQ[i, j] :→
        δaa[f, j, k, i, k] + δa[-ϵ2 b_i f, j, k] +
         δa[ϵ2 b_j f, i, k]
     }]];
```

Definition of *tm*.

```
UU[expr_] // tm[x_, y_, z_] := S[UU[Expand[expr /. {
       a[f_, x, j_] :→ a[f, z, j] + ϵ3 γ[∂_{b_y} f, z, j],
       a[f_, y, j_] :→ a[f, z, j],
       δa[f_, x | y, j_] :→ δa[f, z, j],
       ca[f_, i_, x | y, j_] :→ ca[f, i, z, j],
       δaa[f_, i_, j_, k_, l_] :→
        δaa[f, i // Replace[x | y → z], j,
         k // Replace[x | y → z], l]
      } /. b_{x|y} → b_z]]];
```

Definition of *hm*.

```
UU[expr_] // hm[x_, y_, z_] := S[UU[Expand[expr /. {
       a[f_, i_, x | y] :→ a[f, i, z],
       c[f_, x | y] :→ c[f, z],
       δa[f_, i_, x | y] :→ δa[f, i, z],
       ca[f_, y, j_, x] :→ ca[f, z, j, z] + ϵ4 γ[f, j, z],
       ca[f_, i_, j_, k_] :→
        ca[f, i // Replace[x | y → z], j,
         k // Replace[x | y → z]],
       δaa[f_, i_, y, k_, x] :→ δaa[f, k, z, i, z],
       δaa[f_, i_, j_, k_, l_] :→
        δaa[f, i, j // Replace[x | y → z], k,
         l // Replace[x | y → z]]
      }]]];
```

Definition of *hts*.

```
UU[expr_] // hts[y_, x_] := S[UU[Expand[expr /. {
       a[f_, i_, j_] :→ a[f, i, j] - ϵ5 Kδ_{j,y} γ[∂_{b_x} f, i, y] -
         Kδ_{i,x} Kδ_{j,y} (ϵ6 β[f b_x] + ϵ7 c[f, y] - ϵ8 δβ[b_x ∂_{b_x} f]),
       δa[f_, x, y] :→ δa[f, x, y] - ϵ9 δβ[f b_x],
       ca[f_, i_, j_, k_] :→
        ca[f, i, j, k] + ϵ10 Kδ_{i,y} Kδ_{j,x} γ[f, x, k] +
         Kδ_{j,x} Kδ_{k,y} c[-ϵ11 f b_x, i],
       δaa[f_, i_, j_, k_, l_] :→
        δaa[f, i, j, k, l] + ϵ12 Kδ_{i,x} Kδ_{j,y} δa[-b_x f, k, l] +
         ϵ13 Kδ_{i,x} Kδ_{l,y} (-δa[b_k f, x, j] + δa[b_x f, k, j]) +
         ϵ14 Kδ_{k,x} Kδ_{j,y} (δa[b_i f, x, l] - δa[b_x f, i, l]) +
         ϵ15 Kδ_{k,x} Kδ_{l,y} δa[-b_x f, i, j] +
         ϵ16 Kδ_{i,x} Kδ_{j,l,y} δβ[b_x b_k f] +
         2 ϵ17 Kδ_{x,i,k} Kδ_{y,j,l} δβ[b_x b_x f]
      }]]];
```

Definition of *dm*.

```
dm[x_, y_, z_][expr_] :=
  expr // hts[x, y] // tm[x, y, z] // hm[x, y, z]
```

Renaming operations.

```
tσ[x_List, y_List][expr_] := Module[{r = Thread[x → y]},
   S[expr /. b_i_ :→ b_{i /. r} /. {
      a[f_, i_, j_] :→ a[f, i /. r, j],
      δa[f_, i_, j_] :→ δa[f, i /. r, j],
      ca[f_, i_, j_, k_] :→ ca[f, i, j /. r, k],
      δaa[f_, i_, j_, k_, l_] :→ δaa[f, i /. r, j, k /. r, l]
     }]];
tσ[x_, y_][expr_] := tσ[{x}, {y}][expr];
hσ[x_List, y_List][expr_] :=
  Module[{r = Thread[x → y]},
   S[expr /. {
      a[f_, i_, j_] :→ a[f, i, j /. r],
      c[f_, i_] :→ c[f, i /. r],
      δa[f_, i_, j_] :→ δa[f, i, j /. r],
      ca[f_, i_, j_, k_] :→ ca[f, i /. r, j, k /. r],
      δaa[f_, i_, j_, k_, l_] :→ δaa[f, i, j /. r, k, l /. r]
     }]];
hσ[x_, y_][expr_] := hσ[{x}, {y}][expr];
dσ[x_, y_][expr_] := expr // tσ[x, y] // hσ[x, y];
```

Definition of *tb*.

```
tb[x_][UU[L_], UU[R_]] :=
  Module[{p}, S[UU[Expand[Distribute[p[L, R]] /. {
        p[0, _] → 0, p[_, 0] → 0,
        p[_β | _δβ | _c | _δa | _ca | _δaa,
          _β | _δβ | _c | _δa | _ca | _δaa] → 0,
        p[u_β | u_δβ | u_c | u_δa | u_ca | u_δaa, v_a] :>
          -p[v, u]
       } /. {
       p[a[f_, x, j_], u_] :> (u /. {
            β[g_] :> ϵ18 γ[f ∂_{bx} g, x, j],
            a[g_, k_, l_] :> ϵ19 γa[f ∂_{bx} g, x, j, k, l] +
              Kδ_{x,k} (-γa[ϵ20 g ∂_{bk} f, k, l, x, j] +
                  ca[ϵ21 f g, l, x, j] - ca[ϵ21 f g, j, k, l]),
            _ → 0
          }),
       p[a[f_, j_, k_], a[g_, x, l_]] /; DQ[j, x] :>
        -γa[ϵ22 g ∂_{bx} f, x, l, j, k],
       p[_, _] → 0
      }]]]];
```

Definition of *thb*.

```
thb[x_, y_][UU[L_], UU[R_]] :=
  Module[{p}, S[UU[Expand[Distribute[p[L, R]] /. {
        p[0, _] → 0, p[_, 0] → 0,
        p[_β | _δβ | _c | _δa | _ca | _δaa,
          _β | _δβ | _c | _δa | _ca | _δaa] → 0,
        p[_a, _β | _δβ] → 0,
        p[β[f_], a[g_, i_, j_]] :>
         Kδ_{y,j} γ[ϵ30 g ∂_{bx} f, i, y],
        p[a[f_, i_, j_], a[g_, k_, l_]] :> Kδ_{y,l} (
           γa[ϵ31 g ∂_{bx} f, k, l, i, j] + Kδ_{x,i} (
              γ[-ϵ32 b_k g ∂_{bx} f, i, j] + δa[ϵ33 b_k g ∂_{bx} f,
                 i, j] - δa[ϵ34 b_i g ∂_{bx} f, k, j] - a[
                 ϵ35 b_k f g, i, j] + a[ϵ35 b_i f g, k,
                 j] + ca[ϵ36 f g, j, k, l] - ca[ϵ36 f g,
                 l, k, j])),
        p[a[f_, i_, j_], c[g_, k_]] :>
         -ϵ37 Kδ_{i,x} Kδ_{k,y} γ[f g, i, j],
        p[a[f_, i_, j_], δa[g_, k_, l_]] :>
         ϵ38 Kδ_{x,i} Kδ_{y,l} (-δa[b_k f g, i, j] + δa[b_i f g, k, j]),
        p[a[f_, i_, j_], ca[g_, k_, l_, m_]] :> Kδ_{x,i} (
           -ϵ39 Kδ_{y,k} γa[f g, i, j, l, m] +
            ϵ40 Kδ_{y,m} (-ca[b_l f g, k, i, j] + ca[b_i f g,
                 k, l, j]) - ϵ41 Kδ_{y,k,m} γ[b_l f g, x, j]),
        p[a[f_, i_, j_], δaa[g_, k_, l_, m_, n_]] :> Kδ_{x,i} (
           ϵ42 Kδ_{y,l} (-δaa[b_k f g, i, j, m, n] + δaa[
                 b_i f g, k, j, m, n]) +
            ϵ43 Kδ_{y,n} (-δaa[b_m f g, k, l, i, j] + δaa[
                 b_i f g, k, l, m, j]) +
            ϵ44 Kδ_{y,l,n} (δa[b_x b_m f g, k, j] - δa[b_k b_m f g,
                 x, j])),
        p[_δβ | _c, _a] → 0,
        p[δa[f_, i_, j_], a[g_, k_, l_]] :>
         ϵ45 Kδ_{x,i} Kδ_{y,l} (-δa[b_k f g, i, j] + δa[b_i f g, k, j]),
        p[ca[f_, m_, i_, j_], a[g_, k_, l_]] :>
         ϵ46 Kδ_{x,i} Kδ_{y,l}
          (-ca[b_k f g, m, i, j] + ca[b_i f g, m, k, j]),
        p[δaa[f_, i_, j_, m_, n_], a[g_, k_, l_]] :>
         ϵ47 Kδ_{x,i} Kδ_{y,l} (-δaa[b_k f g, i, j, m, n] +
              δaa[b_i f g, k, j, m, n]) +
          ϵ48 Kδ_{x,m} Kδ_{y,l}
           (-δaa[b_k f g, i, j, m, n] + δaa[b_m f g, i, j, k, n])
       }]]]];
htb[x_, y_][L_UU, R_UU] := -thb[y, x][R, L];
```

Definition of *hb*.

```
hb[y_][UU[L_], UU[R_]] :=
  Module[{p}, S[UU[Expand[Distribute[p[L, R]] /. {
       p[0, _] → 0, p[_, 0] → 0,
       p[_β | _δβ, _] → 0,
       p[_, _β | _δβ] → 0,
       p[_c | _δa | _ca | _δaa, _c | _δa | _ca | _δaa] → 0,
       p[u_c | u_δa | u_ca | u_δaa, v_a] :> -p[v, u]
      } /. {
      p[a[f_, i_, y], u_] :> (u /. {
          a[g_, j_, k_] :>
           ϵ23 Kδ_{y,k} (a[b_j f g, i, y] - a[b_i f g, j, k]),
          c[g_, j_] :> ϵ24 Kδ_{y,j} ɣ[f g, i, j],
          δa[g_, j_, k_] :>
           ϵ25 Kδ_{y,k} (δa[b_j f g, i, y] - δa[b_i f g, j, k]),
          ca[g_, j_, k_, l_] :>
           Kδ_{y,j} ɣa[ϵ26 f g, i, j, k, l] +
            Kδ_{y,l} (ca[ϵ27 b_k f g, j, i, y] -
               ca[ϵ27 b_i f g, j, k, l]),
          δaa[g_, j_, k_, l_, m_] :>
           ϵ28 Kδ_{y,k} (δaa[b_j f g, i, y, l, m] -
               δaa[b_i f g, j, k, l, m]) +
            ϵ29 Kδ_{y,m} (δaa[b_l f g, j, k, i, y] -
               δaa[b_i f g, j, k, l, m])
          }),
      _p → 0
     }]]]];
```

Definition of *db*.

Using $h_1h_2t_1t_2 \to h_1h_2t_1t_2 \to h_1h_2t_2t_1 \to h_2h_1t_2t_1 \to h_2h_1t_2t_1$:

```
db[x_][u_UU, v_UU] := Module[{t, h}, Plus[
    htb[x, x][u // tσ[x, t], v // hσ[x, h]] // tm[t, x, x] //
     hm[x, h, x],
    tb[x][u, v // hσ[x, h]] // hm[x, h, x],
    hb[x][u, v // tσ[x, t]] // tm[t, x, x],
    thb[x, x][u // hσ[x, h], v // tσ[x, t]] //
      tm[t, x, x] // hm[x, h, x]
   ]];
```

The bracket.

```
bb[S_List] := Module[{w, bar, t, n = 0},
   bar[x_] := -x;
   w = #2 // dσ[S, bar /@ S];
   Sum[
    t = db[S⟦k⟧][#1, w // dσ[bar[S⟦k⟧], S⟦k⟧]];
    Do[t = t // dm[bar[S⟦i⟧], S⟦i⟧, S⟦i⟧], {i, 1, k - 1}];
    Do[t = t // dm[S⟦i⟧, bar[S⟦i⟧], S⟦i⟧],
     {i, k + 1, Length@S}];
    t,
    {k, Length@S}
    ]
   ] &
bb[S___] := bb[{S}]
```

AutoAd.

```
AutoAd[B_, x_][y_] :=
  Module[{pows, states, i, s, seq, sh = 5, dseq,
    sf1, sf2, sf, t1, n},
   pows = NestList[B[x, #] &, y, 20];
   states =
    Union[Cases[pows,
      s_β | s_δβ | s_a | s_c | s_δa | s_ca | s_δaa :>
       ReplacePart[s, 1 → _], ∞]];
   UU@Sum[
     seq = Cases[{#}, states⟦i⟧, ∞] & /@ pows;
     seq = Replace[seq, {{_[f_, ___]} :> f, {} → 0}, {1}];
     dseq = Drop[seq, sh];
     If[Union[Length[MonomialList[#]] & /@ dseq] === {1} ∧
       Union[Length[FactorTermsList[#]] & /@ dseq] ===
        {2},
      sf1 = FindSequenceFunction[
        FactorTermsList[#]⟦1⟧ & /@ dseq];
      sf2 = FindSequenceFunction[
        FactorTermsList[#]⟦2⟧ & /@ dseq];
      sf = (sf1[#] sf2[#] &),
      (*Else*)
      sf = FindSequenceFunction[dseq,
        FunctionSpace → {"ConstantRecursive",
          "HolonomicSequence", "Polynomial",
          "RationalFunction", "HypergeometricTerm"}]];
     ReplacePart[states⟦i⟧,
      1 → Simplify[Sum[seq⟦n + 1⟧/n!, {n, 0, sh - 1}] + Sum[sf[n + 1 - sh]/n!, {n, sh, ∞}]]],
     {i, Length@states}]];
(* Hint: Perhaps improve using Variables,
CoefficientList,
FromCoefficientList *)
```

The scattering of a tail by an exponential of tails.

```
AutoAd[bb[1, 2], UU@a[1, 1, 2]][UU@a[1, 1, 0]]
```

$$\mathrm{UU}\Big[a[1,1,0] + ca[\epsilon_5,0,1,2] + ca\Big[-\frac{\left(1-e^{-b_1\epsilon_{10}}\right)\epsilon_5}{b_1\epsilon_{10}},2,1,0\Big] + \delta aa\Big[-\frac{\epsilon_5\left(-1+e^{-b_1\epsilon_{10}}+b_1\epsilon_{10}\right)}{b_1^2\epsilon_{10}},1,0,1,2\Big]\Big]$$

The scattering of a tail by an exponential of heads.

```
AutoAd[bb[1, 2], UU@a[1, 1, 2]][UU@a[1, 2, 0]]
```

$$\mathrm{UU}\Big[a[e^{b_1\epsilon_{10}},2,0] + a\Big[-\frac{\left(-1+e^{b_1\epsilon_{10}}\right)b_2}{b_1},1,0\Big] + c\Big[\frac{b_2\epsilon_5\left(1-e^{b_1\epsilon_{10}}+e^{b_1\epsilon_{10}}b_1\epsilon_{10}\right)}{b_1},0\Big] + ca\Big[\frac{\left(1-e^{-b_1\epsilon_{10}}\right)\epsilon_5}{b_1\epsilon_{10}},2,1,0\Big] +$$
$$ca\Big[-\frac{\epsilon_5\left(1-e^{b_1\epsilon_{10}}+e^{b_1\epsilon_{10}}b_1\epsilon_{10}\right)}{b_1\epsilon_{10}},0,2,2\Big] +$$
$$ca\Big[\frac{\epsilon_5\left(-2\left(-1+e^{b_1\epsilon_{10}}\right)b_2+b_1\left(1-e^{b_1\epsilon_{10}}+\left(1+e^{b_1\epsilon_{10}}\right)b_2\epsilon_{10}\right)\right)}{b_1^2\epsilon_{10}},0,1,2\Big] +$$
$$\delta a\Big[-\frac{b_2\epsilon_5\left(1-e^{b_1\epsilon_{10}}+e^{b_1\epsilon_{10}}b_1\epsilon_{10}\right)}{b_1^2},1,0\Big] +$$
$$\delta aa\Big[\frac{\epsilon_5\left(1-e^{b_1\epsilon_{10}}+e^{b_1\epsilon_{10}}b_1\epsilon_{10}\right)}{b_1^2\epsilon_{10}},1,0,2,2\Big] + \delta aa\Big[-\frac{1}{b_1^3\epsilon_{10}}e^{-b_1\epsilon_{10}}\epsilon_5\left(-2e^{b_1\epsilon_{10}}\left(-1+e^{b_1\epsilon_{10}}\right)b_2+b_1\left(-\left(-1+e^{b_1\epsilon_{10}}\right)^2+e^{b_1\epsilon_{10}}\left(1+e^{b_1\epsilon_{10}}\right)b_2\epsilon_{10}\right)\right),1,0,1,2\Big]\Big]$$

Definition of *ct*.

```
ct[s_] := ct[s, s]; ct[] = ct[0, 0];
ct[h_, t_][UU[L_], UU[R_]] := Module[{p}, S[UU[Distribute[p[L, R]] /. {
      p[_β | _δβ, _] → 0,
      p[a[f_, i_, h], β[g_]] ⧴ β[f b_i ((∂_(b_t) g) /. b_t → 0)],
      p[a[f_, i_, h], a[g_, t, j_]] ⧴ a[f (g /. b_t → 0), i, j],
      p[a[f_, i_, h], a[g_, j_, k_]] ⧴ a[f b_i ((∂_(b_t) g) /. b_t → 0), j, k],
      p[a[f_, i_, h], c[g_, j_]] ⧴ c[f b_i ((∂_(b_t) g) /. b_t → 0), j],
      p[a[f_, i_, h], δa[g_, t, j_]] ⧴ δa[f (g /. b_t → 0), i, j],
      p[a[f_, i_, h], δa[g_, j_, k_]] ⧴ δa[f b_i ((∂_(b_t) g) /. b_t → 0), j, k],
      p[a[f_, i_, h], ca[g_, k_, t, j_]] ⧴ ca[f (g /. b_t → 0), k, i, j],
      p[a[f_, i_, h], ca[g_, l_, j_, k_]] ⧴
       ca[f b_i ((∂_(b_t) g) /. b_t → 0), l, j, k],
      p[a[f_, i_, h], δaa[g_, t, j_, t, k_]] → 0,
      p[a[f_, i_, h], δaa[g_, t, j_, k_, l_]] ⧴
       δaa[f (g /. b_t → 0), i, j, k, l],
      p[a[f_, i_, h], δaa[g_, j_, k_, t, l_]] ⧴
       δaa[f (g /. b_t → 0), j, k, i, l],
      p[a[f_, i_, h], δaa[g_, j_, k_, l_, m_]] ⧴
       δaa[f b_i ((∂_(b_t) g) /. b_t → 0), j, k, l, m],
      p[a[__], _] → 0,
      p[c[f_, h], β[g_]] ⧴ δβ[f ((∂_(b_t) g) /. b_t → 0)],
      p[_c, _β] → 0,
      p[c[f_, h], a[g_, t, j_]] ⧴ c[f (g /. b_t → 0), j],
      p[c[f_, h], a[g_, j_, k_]] ⧴ δa[f ((∂_(b_t) g) /. b_t → 0), j, k],
      p[_c, _a] → 0,
      p[_c | _δa | _ca | _δaa, _δβ | _c | _δa | _ca | _δaa] → 0,
      p[δa[f_, i_, h], β[g_]] ⧴ δβ[f b_i ((∂_(b_t) g) /. b_t → 0)],
      p[δa[f_, i_, h], a[g_, t, j_]] ⧴ δa[f (g /. b_t → 0), i, j],
      p[δa[f_, i_, h], a[g_, j_, k_]] ⧴ δa[f b_i ((∂_(b_t) g) /. b_t → 0), j, k],
      p[_δa, _] → 0,
      p[ca[_, h, _, h], _] → 0,
      p[ca[f_, h, i_, j_], β[g_]] ⧴ δa[f ((∂_(b_t) g) /. b_t → 0), i, j],
      p[ca[f_, i_, j_, h], β[g_]] ⧴ c[f b_j ((∂_(b_t) g) /. b_t → 0), i],
      p[ca[f_, h, i_, j_], a[g_, t, k_]] ⧴ ca[f (g /. b_t → 0), k, i, j],
      p[ca[f_, h, i_, j_], a[g_, k_, l_]] ⧴
       δaa[f ((∂_(b_t) g) /. b_t → 0), i, j, k, l],
      p[ca[f_, i_, j_, h], a[g_, t, k_]] ⧴ ca[f (g /. b_t → 0), i, j, k],
      p[ca[f_, i_, j_, h], a[g_, k_, l_]] ⧴
       ca[f b_j ((∂_(b_t) g) /. b_t → 0), i, k, l],
      p[_ca, _] → 0,
      p[δaa[_, _, h, _, h], _] → 0,
      p[δaa[f_, i_, h, j_, k_], β[g_]] ⧴ δa[f b_i ((∂_(b_t) g) /. b_t → 0), j, k],
      p[δaa[f_, i_, h, j_, k_], a[g_, t, l_]] ⧴
       δaa[f (g /. b_t → 0), i, l, j, k],
      p[δaa[f_, i_, h, j_, k_], a[g_, l_, m_]] ⧴
       δaa[f b_i ((∂_(b_t) g) /. b_t → 0), j, k, l, m],
      p[δaa[f_, i_, j_, k_, h], β[g_]] ⧴ δa[f b_k ((∂_(b_t) g) /. b_t → 0), i, j],
      p[δaa[f_, i_, j_, k_, h], a[g_, t, l_]] ⧴
       δaa[f (g /. b_t → 0), i, j, k, l],
      p[δaa[f_, i_, j_, k_, h], a[g_, l_, m_]] ⧴
       δaa[f b_k ((∂_(b_t) g) /. b_t → 0), i, j, l, m],
      p[_δaa, _] → 0
     }]]];
```